
Abstract – In this paper linear, frequency-dependent VHDL-
AMS model for the Current-Feedback Operational Amplifier 
(CFOA) is introduced. The model simulates the actual perfor-
mance of typical CFOAs for a wide range of frequencies. For 
creating the model, simplification and build-up techniques 
known from macromodelling operational amplifiers have been 
adapted. Model parameters are extracted for the CFOA AD8001 
from Analog Devices as an example. To confirm the validity, the 
simulation results are compared with the manufacturer’s data 
and with the results obtained by simulation of the PSpice macro-
model supplied from Analog Devices. 

I. INTRODUCTION 

The CFOAs, as a variety of operational amplifiers (op 
amps) family, have been realized to overcome the finite gain-
bandwidth product of VFOAs. The CFOAs can be represented 
as a cascade structure of positive second-generation current 
conveyor (CCII+) and an additional voltage buffer. Nowadays 
the CFOAs have emerged as one of the most important active 
building block in developing many analog and mixed-signal 
circuits and systems [1-3]. Testing the workability of the 
developed electronic circuits is usually done using SPICE 
(Simulation Program with Integrated Circuit Emphasis) with a 
company specifying macromodel for the CFOA. A variety of 
SPICE macromodels for the CFOAs, are available in the lite-
rature [4-10]. The majority of published SPICE based macro-
models are presented as device- (transistor-) level models, 
called micromodels. Typically the input stage of the micro-
models is а device-level model and other stages consist of an 
extensive number of passive elements, ideal controlled sour-
ces and ideal diodes. 

Testing a complete analog system via transistor-level simu-
lation is an extremely difficult process and can often become 
infeasible due to the limitation of simulation capacity. A simi-
lar difficulty is encountered when high-level design is perfor-
med for the whole system. For these reasons, compact models 
of analog blocks, and in particular, op amps are desired which 
can be substituted in place of the actual transistor-level netlist 
to speed up the simulation without sacrificing any of the 
required accuracy. 

One method to decrease simulation time and improve the 
convergence, without a significant loose of information, is by 
using behavioural modelling technique. Behavioural mo-

delling is a way of providing macroscopic models of the 
corresponding microscopic (microelectronic) circuits. The use 
of behavioural modelling for op amps has been well known 
for a few years. Behavioural models are realized by using 
structural macromodelling, the C code modelling, Analog Be-
havioural Modelling (ABM) available in PSpice A/D simula-
tors and finally the behavioural modelling with VHDL-AMS. 

Nowadays one of the most effective techniques for beha-
vioural modelling of analog and mixed electronic circuits is 
by using VHDL-AMS. The conveyed literature survey reveals 
that behavioural models for the CFOAs have not been 
available yet. Without any doubt behavioural models for basic 
types of CFOAs, are necessary for simulating complex 
circuits. The goal of this paper therefore is to develop a simple 
behavioural VHDL-AMS model that accurately simulates the 
frequency performance of most common CFOAs.   

II. CFOAS 

The most common CFOA is equivalent to a CCII+ plus an 
output voltage buffer. These op amps have a high impedance 
non-inverting input y , a low-impedance inverting input x , a 
current output z  and the voltage output о. In some of the 
CFOAs the port z , between the first stage (CCII+) and the 
second stage (voltage buffer), is defined as an external pin. 
This allows the usage of CFOAs in some specific selective 
amplifier circuits, sinusoidal oscillators and multivibrators. 
The port о is the output of the voltage buffer, where the 
resistance is very low (magnitude of several ohms). The 
simplified model of the CFOA is presented in Fig. 1. This mo-
del contains three parameters to be determined by measu-

rements, i.e. yxα , the y  to x  voltage gain, xzα , the x  to z  

voltage gain and zoα , the z  to o  voltage gain. All these 
parameters are ideally equal to 1. For this model the general 
relation between input and output voltages and currents can be 
given as 0=yi , yyxx vv α= , xxzz ii α=  and ozoz vv α= . If 

pin z  is defined as an external port yxα  is measured by app-

Fig. 1. A simple model of the real CFOA. 

lying a voltage signal at port y  and measuring the resulting 
voltage at port x . xzα is measured by injecting a current 
signal at port x  and measuring the output current at port z . 
Whereas zoα  is measured the same way as yxα . 

III. BEHAVIOURAL MODELLING WITH VHDL-AMS 

The technical requirement for effective models is generally 
agreed when the simplest model possible is developed. Simple 
models have a number of advantages. They can be developed 
faster, are more flexible, require less data, run faster and it is 
easier to interpret the results since the structure of the model is 
better understood. As the complexity increases, these advan-
tages are lost [11]. The proposed behavioural model of the 
CFOA is developed following the design method based on a 
Top-Down analysis approach and applying simplification and 
build-up technique. The process of model building and testing 
can be broken down into three basic steps: structure the 
model, build the model and validate the model [12]. 

A. A behavioural language: VHDL-AMS 

VHDL-AMS is a comparatively new standard 1076.1 of 
VHDL that support hierarchical description and simulation of 
analog, digital and mixed-signal applications with conserva-
tive and non-conservative equations [13, 14]. On the analog 
side, a variety of abstraction levels is supported. The VHDL-
AMS modelling is not restricted to electronic circuits, but also 
supports thermal, mechatronic, optical and other systems. 

  B. A CFOA behavioural VHDL-AMS model 

The behavioural model of CFOA is built using the electrical 
characteristic curves obtained by analyses of the PSpice mac-
romodel for AD8001AN [15]. In fact, the monolithic op amp 
AD8001AN is a typical representative of the commercially 
available CFOAs. The circuit diagram of the new CFOA mo-
del is shown in Fig. 2, where the different stages are presented 
with controlled sources and passive components. 

The proposed model including effects such as (1) accurate 
inverting and non-inverting input impedance, (2) offset vol-
tage and input bias currents, (3) dc transfer characteristic, (4) 
ac small-signal frequency response, (5) transient response un-
der a wide range of conditions, (6) input and output voltage 
and current limitation and (7) output resistance. 

 The mathematical equations that describe the model can be 
given as 

where ioyy vvv +=1  and pω  is the dB3−  radian frequency 
for the dominant pole of a typical CFOA. 

Short-circuit current limiting is simulated with model para-
meters pimax_  and nimax_ . In short-circuit mode the output 

current roi  is limited to one of the values pimax_  or nimax_ .  
Fig. 3 shows the behavioural VHDL-AMS model of CFOA. 

The library clause and the use clause make all declarations in 
the packages electrical_systems and math_real vi-
sible in the model. This is necessary, because the model uses 
nature electrical from package electrical_system 

Fig. 2. Circuit diagram of the Current-Feedback Operational Amplifier (CFOA) behavioural model. 

and constant math_2_pi for the value of π  from package 
math_real. The proposed CFOA model is composed by an 
entity and an architecture, where bold text indicates reserved 
words and upper-case text indicates predefined concepts. The 
entity declares the generic model parameters and specifies 
three interface terminals of nature electrical. The para-
meters are given with concrete numerical values for the 
CFOA AD8001AN. The proposed CFOA model includes the 
following electrical terminals: non-inverting input – y, 
inverting input – x and output – o (or output). Furthermore, 
the model has three inner terminals: n0, n1 and z. They are 
used to specify the voltages vy1, vz and vo1, respectively. 

 The architecture of the model is subdivided into three 
parts: input stage, transfer stage and output stage. It contains 
the implementation of the model. The architecture is coded by 
using a style combining structural and behavioural elements. 
The structural description is the netlist of the model and the 
behavioural description consists of simultaneous statements to 
describe continuous behaviour.  

IV. MODEL PERFORMANCE 

The verification of the proposed behavioural model, 
shown Fig. 3, is performed by comparing simulation results 
with the manufacturer’s data and with the results obtained by 

simulation of the PSpice macromodel supplied from Analog 
Devices. The test circuits for simulation modelling are crea-
ted following the test conditions given in the semiconductor 
data books of the corresponding CFOA. The power supply 
voltages of the circuits are chosen to be ±5V. The model 
parameters yxα , xzα  and zoα  are approximately equal to 1. 
The simulation modelling of the proposed behavioural model 
of the CFOA was implemented within simulation program 
System Vision 5.5 (from Mentor Graphics) and simulation 
modelling of the macromodel AD8001AN/AD was perfor-
med with OrCAD PSpice. 

In Table 1 are presented the simulation results and the data 
sheet parameters for the CFOA AD8001AN. As a result, the 
proposed behavioural CFOA model corresponds with the 
parameters of the AD8001AN and it is more accurate at DC 
and low frequencies than the AD8001AN/AD. 

TABLE I 
COMPARISON BETWEEN SIMULATION RESULTS AND DATA 

SHEET 

Parameter AD8001AN 
Data sheet 

AD8001AN/AD 
macromodel 

The proposed 
CFOA model 

DC parameters 
Input offset 
voltage iov  mV2  mV5  mV2  

− Input bias 
current − xibi _  Aμ5  Aμ6  Aμ01,5  

+ Input bias 
current − yibi _  Aμ3  Aμ25  Aμ3  

AC parameters 
Transresistan-
ce − zr  Ωk900  Ωk316  Ωk900  

Input resistance: 
+Input − yr  ΩM10  ΩM3,3  ΩM10  
−Input − xr  Ω50  Ω51  Ω50  
+ Input capa-
citance − yC  pF5,1  pF5,1  pF5,1  

–3 dB Small 
Signal Band-
width 

MHz880 (1) 
MHz260 (2) 

MHz370 (1) 
MHz427 (2) 

MHz870 (1) 
MHz270 (2) 

Dominant pole 
frequency− pf  kHz700  kHz896  kHz01,700  

Slope (open 
loop) decdB /20−  decdB /20−  decdB /20−  

Output voltage 
swing − maxV  V1,3± (3) V68,2± (3) V09,3± (3) 

Short circuit 
current − maxI  mA100  mA64,85  mA01,100  

Output resis-
tance − or  Ω50  Ω51  Ω50  

Notes: (1) 1+=UA  ( Ω= 649FR  [15]); (2) 10+=UA  ( Ω= 470FR  and 

Ω= 51NR  [15]); (3) Ω=100LR . 

To validate the proposed CFOA model, simulation of De-
liyannis-Frend biquad bandpass filter shown in Fig. 3 was 
carried out for both the aforementioned model and the macro-
model given by Analog Devices in the PSpice library. The 
results are shown in Fig. 4a and Fig. 4b. 

library IEEE; 
use IEEE.math_real.all; 
use IEEE.electrical_systems.all; 
entity CFOA is 
generic ( 
  iib   : CURRENT := -2.0e-6;  --input bias current 
  iio   : CURRENT := 4.0e-6;  --input offset current 
  vio   : VOLTAGE := 2.0e-3;  --input offset voltage 
  ry   : RESISTANCE := 10.0e6;  --positive input resistance 
  cy   : CAPACITANCE := 1.5e-12;  --positive input capacitance 
  rx   : RESISTANCE := 50.0;  --negative input resistance 
  rz   : RESISTANCE := 900.0e3; --transresistance 
  ro   : RESISTANCE := 50.0;  --output resistance 
  ayx   : REAL := 1.0;    --the Y to X voltage gain 
  axz   : REAL:= 1.0;   --the X to Z current gain 
  fp   : REAL:= 700.0e3;   --fp of gain 
  azo   : REAL:= 1.0;   --the Z to O voltage gain 
  v_max_p : VOLTAGE := 5.0;  --max pos. output voltage 
  v_max_n : VOLTAGE := -5.0;  --max neg. output voltage 
  i_max_p : CURRENT := 110.0e-3; --max pos. output current 
  i_max_n : CURRENT := -110.0e-3); --max neg. output current 
  port (terminal y, x, output : electrical); 
end entity CFOA; 
architecture arch_CFOA of CFOA is 
-- inner terminals 
  terminal n0, n1, z: electrical; 
-- inner branch quantities and free quantities 
  quantity vy across        y to electrical_ref; 
  quantity v_io across i2 through y to n0; 
  quantity vy1 across iy, icy, iib_y through n0 to electrical_ref; 
  quantity vx across ix, iib_x through x to electrical_ref; 
  quantity iz through electrical_ref to z; 
  quantity irz through z to electrical_ref; 
  quantity vz across   z to electrical_ref; 
  quantity vo1 across io1 through n1 to electrical_ref; 
  quantity vro across iro through n1 to output; 
  quantity voutput across output to electrical_ref; 
  quantity iro_h : current; 
-- constants 
  constant wp: REAL := fp * math_2_pi; -- -3db frequency in radians 
  constant num: REAL_VECTOR := (0 => axz); 
  constant den: REAL_VECTOR := (1.0, 1.0/wp); 
begin 
--***input stage***-- 
  v_io == vio; 
  iy  == vy1/ry; 
  icy == cy * vy1'dot; 
  iib_y == iib + iio/2.0; 
  vx == ayx * vy1 + ix * rx; 
  iib_x == iib - iio/2.0; 
--***transfer stage***-- 
  iz == ix'ltf(num, den); 
  irz == vz/rz; 
--***output stage***-- 
  iro_h == vro/ro; 
--limitation of the output voltage 
  if vz'above(v_max_p) use vo1 == v_max_p; 
 elsif not vz'above(v_max_n) use vo1 == v_max_n; 
    else vo1 == azo * vz; 
end use; 
--limitation of the output current 
if iro_h'above(i_max_p) use iro == i_max_p; 
 elsif not iro_h'above(i_max_n) use iro == i_max_n; 
    else iro == iro_h; 
 end use; 
end architecture arch_CFOA; 

Fig. 3. A CFOA behavioural VHDL-AMS model.  

Fig. 3. The Deliyannis-Frend biquad filter used in testing the 

different CFOA models. 
The result of the simulations performed shows that the pre-

sented model corresponds to the AD8001AN/AD for frequen-
cies from 0 up to MHz100 . At higher frequencies the macro-
model supplied from Analog Devices is superior to the beha-
vioural one, with the expense of a much more complicated set 
of mathematical equations that describe the proposed model. 

The purpose of developing that CFOA model was to intro-
duce a behavioural model that is suitable for analyses of the 
mixed-signal applications with VHDL-AMS simulators yet 
quite accurate to a wide frequency range. 

 
Fig. 4a. The simulated response of the bandpass filter shown in 

Fig. 3 for the AD8001AN/AD. 

 
Fig. 4b. The simulated response of the bandpass filter shown in 

Fig. 3 for the proposed CFOA model. 

V. CONCLUSION 

In this paper behavioural VHDL-AMS model of CFOA has 
been presented. The proposed model accurately describes the 
behaviour of a typical CFOA over a wide frequency range, 
including electrical parameters such as inverting and non-in-
verting input impedance, open-loop transresistance, band-
width, voltage and current limitations, output resistance, etc. 
To achieve simplicity of the mathematical equations desc-
ribing the model, it neglects several second-order effects 
found in the CFOAs (such as the frequency dependence of the 
output impedance of the voltage followers, the CMRR and the 
temperature effects). One of the aims of the further work is to 
explore the possibility of modelling second-order effects of 
the typical CFOA, which are taken into consideration for 
designing high-frequency mixed-signal applications. 

